DESIGN OF DIGITAL CIRCUITS AND SYSTEMS

ASM with Datapath II

Instructor: Justin Hsia

Teaching Assistants:

Colton Carroll
Hemil Patel
Rasya Fawwaz
Grace Zhou
Quinlyn Donohue
Rose Maresh

Relevant Course Information

- Homework 3 due Friday (4/24)
- Homework 4 released Thursday

- Quiz 2 (ROM, RAM, Reg files) this Thursday at 11:**50** AM
  - Based heavily on Homework 2
  - Memory sizing, addressing, initialization, and implementation (*i.e.*, circuit diagram)

- Lab 3 reports due next Friday (5/1)
  - Ideally finish by early next week so you can start Lab 4, which will be released this Thursday

ASMD Design Procedure

- From problem description or algorithm pseudocode:
  1) **Identify necessary datapath components and operations**
  2) **Identify states and signals that cause state transitions** (external inputs and status signals), based on the necessary sequencing of operations
  3) **Name the control signals** that are generated by the controller that cause the indicated operations in the datapath unit
  4) **Form an ASM chart for your controller**, using states, decision boxes, and signals determined above
  5) **Add the datapath RTL operations** associated with each control signal

Design Example #1



- System specification:
  - Flip-flops E and F
  - 4-bit binary up-counter $A = 0\text{b}A_3A_2A_1A_0$
  - Active-low reset signal $reset\_b$ puts us in state S_idle, where we remain while signal $Start = 0$
  - $Start = 1$ initiates the system's operation by clearing A and F. At each subsequent clock pulse, the counter is incremented by 1 until the operations stop.
  - Bits A_2 and A_3 determine the sequence of operations:
    - If $A_2 = 0$, set E to 0 and the count continues
    - If $A_2 = 1$, set E to 1; additionally, if $A_3 = 0$, the count continues, otherwise, wait one clock pulse to set F to 1 and stop counting (*i.e.*, back to S_idle)

Design Example #1: ASMD Chart

- Synchronous or asynchronous reset?



for sychronous reset, add decision box on reset_b out of every state box:



Design Example #1: Controller Code

```
module controller (set_E, clr_E, set_F, clr_A_F,
                   incr_A, A2, A3, Start, clk,
                   reset_b);

   // port definitions
   input  logic Start, clk, reset_b, A2, A3;
   output logic set_E, clr_E, set_F, clr_A_F, incr_A;

   // define state names and variables
   enum logic [1:0] {S_idle, S_1, S_2 = 3} ps, ns;

   // controller logic w/synchronous reset
   always_ff @(posedge clk)
      if (~reset_b)
         ps <= S_idle;
      else
         ps <= ns;
```

```
   // next state logic
   always_comb
      case (ps)
         S_idle: ns = Start ? S_1 : S_idle;
         S_1:    ns = (A2 & A3) ? S_2 : S_1;
         S_2:    ns = S_idle;
      endcase

   // output assignments
   assign set_E =   (ps == S_1) & A2;
   assign clr_E =   (ps == S_1) & ~A2;
   assign set_F =   (ps == S_2);
   assign clr_A_F = (ps == S_idle) & Start;
   assign incr_A =  (ps == S_1);

endmodule  // controller
```

Design Example #1: Datapath Code

```
module datapath (A, E, F, clk, set_E, clr_E, set_F, clr_A_F,
                 incr_A);

   // port definitions
   output logic [3:0] A;
   output logic E, F;
   input  logic clk, set_E, clr_E, set_F, clr_A_F, incr_A;

   // datapath logic
   always_ff @(posedge clk) begin
      if (clr_E)       E <= 1'b0;
      else if (set_E)  E <= 1'b1;
      if (clr_A_F)
         begin
            A <= 4'b0;
            F <= 1'b0;
         end
      else if (set_F)  F <= 1'b1;
      else if (incr_A) A <= A + 4'h1;
   end  // always_ff

endmodule  // datapath
```

Design Example #1: Top-Level Code

```
module top_level (A, E, F, clk, Start, reset_b);

   // port definitions
   output logic [3:0] A;
   output logic E, F;
   input  logic clk, Start, reset_b;

   // internal signals
   logic set_E, clr_E, set_F, clr_A_F, incr_A;

   // instantiate controller and datapath
   controller c_unit (.set_E, .clr_E, .set_F,
                      .clr_A_F, .incr_A, .A2(A[2]),
                      .A3(A[3]), .Start, .clk,
                      .reset_b);
   datapath d_unit (.*);

endmodule  // top_level
```

Design Example #2: Fibonacci

- Design a sequential Fibonacci number circuit with the following properties:



  - `i` is the desired sequence number
  - `f` is the computed Fibonacci number:

$$fib(i) = \begin{cases} 0, & i = 0 \\ 1, & i = 1 \\ fib(i-1) + fib(i-2), & i > 1 \end{cases}$$

  - `ready` means the circuit is idle and ready for new input
  - `start` signals the beginning of a new computation
  - `done_tick` is asserted for 1 cycle when the computation is complete

There are many valid designs. One is shown; some variants will be listed.

Design Example #2: Pseudocode

```
if (i== 0) return 0;          // i_eq_0, zero_f
fm1 = 1;                      // init
fm2 = 0;
for n from 2 to i-1 (inclusive):   // n_lt_i
    temp = fm2;               // iterate
    fm2 = fm1;
    fm1 = temp + fm2;
return fm1 + fm2;
```

Some Variants:

- could have explicitly shown i==1 base case
- any loop bounds that execute i-2 times will work
- could have explicitly used a third f = fm1 + fm2 variable

- Pseudocode analysis:
  - Variables are part of **datapath**; assignments become RTL operations
  - Chunks of related actions should be triggered by **control signals**
  - Decision points become **status signals**

Design Example #2: Block Diagram (Control-Datapath)



Design Example #2: ASMD Chart



Some Variants:

- i == 0 check could be in S_idle
- n could count down
- done_tick could be a Mealy output from S_iter (shows up 1 cycle earlier)
- f could be a separate register from fm1 & fm2 and be updated just once

Design Example #2: Code Outlines

```
fib_control:
   // port definitions
   // define state names and variables
   // controller logic w/synchronous reset
   // next state logic
   // output assignments

fib_datapath:
   // port definitions
   // datapath logic

fib:
   // port definitions
   // define status and control signals
   // instantiate control and datapath
```

Other Hardware Algorithms We Will Look At

- Sequential binary multiplier or divider
- Arithmetic mean

- Lab 4: Bit counting
- Lab 4: Binary search
- Lab 5: Bresenham's line

TECHNOLOGY

BREAK

Hardware Acceleration

- ASMD as a design process can be used to implement software algorithms
- Custom hardware can accelerate operation:
  - Hardware can better exploit parallelism
  - Hardware can implement more specialized operations
  - Hardware can reduce "processor overhead" (*e.g.*, instruction fetch, decoding)
- "Hardware accelerators" are frequently used to complement processors to speed up common, computationally-intensive tasks
  - *e.g.*, encryption, machine vision, cryptocurrency mining

Binary Multiplication

- Multiplication of unsigned numbers:



Parallel Binary Multiplier

- *Parallel* multipliers require a lot of hardware:



Sequential Binary Multiplier (1/2)

- Design a *sequential* multiplier that uses **only one adder and a shift register**
  - Assume one clock cycle to shift and one clock cycle to add
  - More efficient in hardware, less efficient in time

- Considerations:
  - n-bit multiplicand and multiplier yield a product at most how wide?

    2 n-bits wide
  - What are the ports for an n-bit adder?

    A, B, carry-in, carry-out, S

    (n+1)-bit sum
  - How many shift-and-adds do we do and how do we know when to stop?

    n operations, use a counter to track

Sequential Binary Multiplier (2/2)

- Design a *sequential* multiplier that uses **only one adder and a shift register**
  - Assume one clock cycle to shift and one clock cycle to add
  - More efficient in hardware, less efficient in time

- Implementation Notes:
  - If current bit of multiplier is `0`, then skip the adding step (don't bother adding 0)
  - Instead of shifting multiplicand to the left, we will shift the partial sum (and the multiplier) *to the right* (same effect, but hardware is different)
  - We will re-use the multiplier register for the lower half of the product
    - Treat carry, partial sum, and multiplier as one shift register {C, A, Q}

(can discard multiplier bits we've already looked at)

Sequential Binary Multiplier Operation

- A few steps of:

```
   11010111
x  00010111
```



Operation (completed)	C	A	Q	P
Initialize computation	0	00000000	00010111	1000
Add (Q[0] = 1)	0	11010111	00010111	1000
Shift	0	01101011	10001011	0111
Add (Q[0] = 1)	1	01000010	10001011	0111
Shift	0	10100001	01000101	0110
Add (Q[0] = 1)	1	01111000	01000101	0110
Shift	0	10111100	00100010	0101

Binary Multiplier: Specification

- Datapath
  - $(2n+1)$-bit *shift register* with bits split into 1-bit C, n-bit A, and n-bit Q
  - Multiplicand stored in register B, multiplier stored in Q
  - An n-bit *parallel adder* adds the contents of B to A and outputs to $\{C, A\}$
  - A $\lceil \log_2(n+1) \rceil$-bit *counter* P

{C, A, Q} ← {C, A, Q} >> 1

- Control
  - Inputs $Start$ and $Reset$, outputs $Ready$ and $Done$
  - Status signals: Q[0] (or all of Q), P_zero (or all of P)
  - Control signals: Shift_regs, Load_regs (Initialize), Add_regs, Decr_P

Binary Multiplier: Block Diagram



Binary Multiplier: ASMD Chart





Binary Multiplier Implementation

- Controller Logic

$$Load\_regs = S\_idle \cdot Start$$

$$Shift\_regs = S\_shift$$

$$Add\_regs = S\_add \cdot Q[0]$$

$$Decr\_P = S\_add$$

$$Ready = S\_idle$$

$$Done = S\_done$$

Binary Multiplier: Datapath Outline

```
module datapath #(parameter WIDTH=8)
                 (product, Q, P, multiplicand, multiplier, clk,
                  Load_regs, Shift_regs, Add_regs, Decr_P);

   // port definitions
   output logic [2*WIDTH-1:0] product;
   output logic [WIDTH-1:0] Q, P;  // note: unnecessary bits for P
   input  logic [WIDTH-1:0] multiplicand, multiplier;
   input  logic clk, Load_regs, Shift_regs, Add_regs, Decr_P;

   // internal logic
   logic C;
   logic [WIDTH-1:0] A, B;

   // datapath logic









endmodule
```

Binary Multiplier: Datapath Code

```
module datapath #(parameter WIDTH=8)
                 (product, Q, P, multiplicand, multiplier, clk,
                  Load_regs, Shift_regs, Add_regs, Decr_P);

   // port definitions
   ...

   // internal logic
   ...

   // datapath logic
   always_ff @(posedge clk) begin
      if (Load_regs) begin
         A <= 0;  C <= 0;  P <= WIDTH;
         B <= multiplicand;
         Q <= multiplier;
      end
      if (Decr_P)            P <= P - 1;
      if (Add_regs)          {C, A} <= A + B;
      else if (Shift_regs)   {C, A, Q} <= {C, A, Q} >> 1;
   end  // always_ff

   assign product = {A, Q};

endmodule
```